** Clearing Stata memory
capture log close
clear all
set more off, perm
set seed 1234

///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//////////////////////////// Table O.16: Priority Subjects and Gender Performance Gap  (All P1 Scores X Subject FE) ///////
///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

** Opening Phase 2 norm_scores dataset 
use "Work Data/Gender_Phase2_long.dta",clear

*** Creating variables
encode subject, gen (sub)
tab subject, gen (d_sub)
label var sub "Subject"

** Subject dummies
rename d_sub1 Biology
rename d_sub2 Chemistry
rename d_sub3 Geography
rename d_sub4 History
rename d_sub5 Language
rename d_sub6 Mathematics
rename d_sub7 Physics
rename d_sub8 Portuguese
* Labels
label var Biology "Biology"
label var Chemistry "Chemistry"
label var Geography "Geography"
label var History "History"
label var Math "Mathematics"
label var Physics "Physics"
label var Portuguese "Portuguese"
label var Language "Foreign Language"

** Interaction: priority X female
gen fem_priority=female*priority
label var fem_priority "Female $\times$ Priority"

** Interaction: priority X subject
foreach v of varlist Biology-Portuguese {
gen fem_`v'=`v'*female
label var fem_`v' "Female $\times$ `v'"
gen prio_`v'=priority*`v'
label var prio_`v' "Priority $\times$ `v'"
gen fem_prio_`v'=fem_priority*`v'
label var fem_prio_`v' "Female $\times$ Priority $\times$ `v'"
}

global subject "Chemistry Geography History Mathematics Physics"
global subject_fem "fem_Chemistry fem_Geography fem_History fem_Mathematics fem_Physics"

** P1 scores: P1 normalized subject-specific scores
forvalues i=2(1)4 {
gen norm_p1score`i'=norm_p1score^`i'
sum norm_p1score`i'
}

*********************************************************************************
****************   Relative performances ****************************************
*********************************************************************************

******************************** ENEM ********************************

foreach v in norm_enem_w {
bys year female: egen `v'_ave_g=mean(`v')
gen `v'_g=`v'-`v'_ave_g
bys year female: sum `v'_g
}
drop norm_enem_w_ave_g

* Interaction: subject X ENEM
foreach v of varlist Biology-Portuguese {
gen enem_`v'=`v'*norm_enem_w_g
label var enem_`v' "ENEM $\times$ `v'"
gen fem_enem_`v'=female*norm_enem_w_g*`v'
label var fem_enem_`v' "Female $\times$ ENEM $\times$ `v'"
forvalues i=2(1)4 {
gen enem_`v'_`i'=enem_`v'^`i'
gen fem_enem_`v'_`i'=fem_enem_`v'^`i'
}
sum enem_`v'* fem_enem_`v'*
}


global g_pol_enem_sub "enem_Chemistry* enem_Geography* enem_History* enem_Mathematics* enem_Physics*"
d $g_pol_enem_sub

* Priority x relative performance in ENEM:
foreach v in norm_enem_w {
gen `v'_priority_g=`v'_g*priority
forvalues i=2(1)4 {
gen `v'_priority_g`i'=`v'_g^`i'*priority
sum `v'_priority_g`i'
}
}

global g_norm_enem_w_prio norm_enem_w_priority_g*
d $g_norm_enem_w_prio

******************************** Phase 1 scores ********************************

foreach v in norm_p1score {

tab year, sum(`v')
bys year subject female: egen gs_`v'_ave=mean(`v')
gen gs_`v'=`v'-gs_`v'_ave
bys year female subject: sum gs_`v'
drop gs_`v'_ave

forvalues i=2(1)4 {
gen gs_`v'`i'=gs_`v'^`i'
sum gs_`v'`i'
}

global gs_pol_`v' gs_`v' gs_`v'2 gs_`v'3 gs_`v'4
d $gs_pol_`v'

* Priority x Phase 1 scores:
gen gs_`v'_prio=gs_`v'*priority
forvalues i=2(1)4 {
gen gs_`v'_prio`i'=gs_`v'`i'*priority
sum gs_`v'_prio*
}
}

global gs_pol_norm_p1score_prio gs_norm_p1score_prio*
d $gs_pol_norm_p1score_prio

*********************************************************************************
**************** Main sample ****************************************************
*********************************************************************************

* 1) Only years before the affirmative action took place
drop if aa_year==1
tab year
drop if year==2000
tab year

* 2) Drop Portuguese and Foreign Language (in Phase 1 there is no Portuguese or Foreign Language exams - For Portuguese Phase 1 has an essay)
 tab subject, sum(norm_p1score)
 drop if subject=="lang" | subject=="port" 
 tab subject, sum(norm_p1score)
 drop Language Portuguese prio_Language prio_Portuguese fem_prio_Language fem_prio_Portuguese 

******************** Phase 1 scores, All subjects ********************

sort inscri2
order subject gs_norm_p1score 
label var gs_norm_p1score "Phase 1 scores"

local j=1	
levelsof subject, local(levels) 
foreach i of local levels {

	by inscri2: egen gs_norm_p1score_`i' = mean(gs_norm_p1score) if subject=="`i'" // relative performance in phase 1, similar to main specification
	by inscri2: egen aux_`i' = max(gs_norm_p1score_`i') 
	replace gs_norm_p1score_`i' = aux_`i' 
	drop aux_`i' 
	
	if `j' ==1 loc title= "Bio"
	if `j' ==2 loc title= "Chem"
	if `j' ==3 loc title= "Geog"
	if `j' ==4 loc title= "Hist"
	if `j' ==5 loc title= "EngFr"
	if `j' ==6 loc title= "Math"
	if `j' ==7 loc title= "Phys"
		
	label var gs_norm_p1score_`i' "P1 scores `title' "

	local ++j
}

global gs_norm_p1score_all gs_norm_p1score_biol gs_norm_p1score_chem gs_norm_p1score_geog gs_norm_p1score_hist gs_norm_p1score_math gs_norm_p1score_phy 
d $gs_norm_p1score_all

rename Mathematics Math

* Interaction: subject X  Phase 1 scores
foreach j of varlist Biology-Physics {
levelsof subject, local(levels) 
foreach i of local levels {
gen gs_norm_p1score_`i'_`j'=`j'*gs_norm_p1score_`i'
label var  gs_norm_p1score_`i'_`j' "Phase 1 scores `i' $\times$ `j'"
}
}

foreach j of varlist Biology-Physics {
levelsof subject, local(levels) 
foreach i of local levels {
forvalues n=2(1)4 {
gen gs_norm_p1score_`i'_`j'`n'=(gs_norm_p1score_`i'^`n')*`j'
sum gs_norm_p1score_`i'_`j'`n'
}
}
}

d  gs_norm_p1score_biol_*  gs_norm_p1score_chem_* gs_norm_p1score_geog_*  gs_norm_p1score_hist_* gs_norm_p1score_math_*  gs_norm_p1score_phy_* 
 
global gs_pol_norm_p1score_all_sub  gs_norm_p1score_biol_*  gs_norm_p1score_chem_* gs_norm_p1score_geog_*  gs_norm_p1score_hist_* gs_norm_p1score_math_*  gs_norm_p1score_phy_* 
d $gs_pol_norm_p1score_all_sub

gen Mathematics=Math

*********************************************************************************
****************   Regressions **************************************************
*********************************************************************************

** Labeling variables
label var priority "Priority"
label var female "Female"
label var norm_enem_w_g "ENEM"

estimates clear

reg norm_score female priority fem_priority norm_enem_w_g , cluster(inscri2) 
estimates store reg1
reg norm_score female priority fem_priority norm_enem_w_g $subject $subject_fem , cluster(inscri2) 
estimates store reg2
reghdfe norm_score priority fem_priority $subject $subject_fem, cluster(inscri2) absorb(inscri2)
estimates store reg3
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub, cluster(inscri2) absorb(inscri2)
estimates store reg4
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio, cluster(inscri2) absorb(inscri2)
estimates store reg5
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score_all_sub , cluster(inscri2) absorb(inscri2)
estimates store reg6
reghdfe norm_score priority fem_priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score_all_sub $gs_pol_norm_p1score_prio, cluster(inscri2) absorb(inscri2)
estimates store reg7

estadd local sub_fe "No":  reg1
estadd local sub_fe "Yes": reg2 reg3 reg4 reg5 reg6 reg7

estadd local subgender_fe "No":  reg1
estadd local subgender_fe "Yes": reg2 reg3 reg4 reg5 reg6 reg7

estadd local ind_fe "No": reg1 reg2 
estadd local ind_fe "Yes": reg3 reg4 reg5 reg6 reg7

estadd local enemsub "No":  reg1 reg2 reg3 
estadd local enemsub "Yes": reg4 reg5 reg6 reg7

estadd local enemprio_pol4 "No":  reg1 reg2 reg3 reg4 
estadd local enemprio_pol4 "Yes": reg5 reg6 reg7

estadd local p1score_all_sub "No": reg1 reg2 reg3 reg4 reg5
estadd local p1score_all_sub "Yes": reg6 reg7

estadd local p1scoreprio_pol4 "No": reg1 reg2 reg3 reg4 reg5 reg6
estadd local p1scoreprio_pol4 "Yes": reg7

* Tex
esttab reg1 reg2 reg3 reg4 reg5 reg6 reg7 using "Output/p_Robustness_Interactions_P1allPol_Sub.tex", se star(* 0.10 ** 0.05 *** 0.01) nogap ///
stats(r2_a N N_clust sep sub_fe subgender_fe ind_fe enemsub  enemprio_pol4 p1score_all_sub p1scoreprio_pol4, fmt(%9.3fc %9.0fc %9.0fc %1s %3s %3s %3s %3s %3s %3s %3s %3s %3s %3s) labels("$\bar{R}^2$" "Number of observations"  "Number of applicants" " " "Subject FE" "Subject-gender FE" "Individual FE" "ENEM $\times$ Subject FE" "ENEM $\times$ Priority" "Phase 1 scores (all subjects)  $\times$ Subject FE" "Phase 1 scores $\times$ Priority")) b(%7.3f) se(%7.3f)  booktabs replace f label nomtitle collabels(none) keep(female priority fem_priority norm_enem_w_g) refcat(female " \\ \multicolumn{7}{l}{\textit{Dependent variable: Phase 2 normalized subject-specific scores}} \\", nolabel)